Allgemeines zu isql und Spezielles zur Formularentwicklung bzw. zur Eingabe von Konferenzbänden in CSLIB 2000

Gruppe 6 Noyan Ilmen Nicolai Leymann Markus Rebensburg

TU Berlin Fachbereich Informatik Fachgebiet Wissensbasierte Systeme Dr.-Ing. Ulrike Reiner

Berlin, den 13. August 1996

Inhaltsverzeichnis

0 Vorbemerkung 2

1 Basis-Formularerzeugung 2 1.1 Methoden . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 2 1.1.1 sformbld . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 2 1.1.2 isql-Formulargenerator . . . . . . . . . . . . . . . . . . . . . . . . . 3 1.1.3 Handarbeit . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 3 1.2 Zusammenfassung . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 4

2 isql-Formularspezialitäten 5

3 Während der Eingabe 6 3.1 Beispielformular . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 7

4 Konferenzbandformular (Souce-code) 9

5 Konstanten 10

Tabellenverzeichnis

1 Konstanten . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 10

0 Vorbemerkung

Damit überhaupt etwas funktioniert, muß das Environment angepasst werden. Meines sieht so aus:

setenv PATH ${PATH}\:/home/wbs/ul/cslib-marzena/informix5-ole/bin
setenv PATH ${PATH}\:/home/wbs/ul/cslib-marzena/informix5-ole/etc
setenv INFORMIXDIR /home/wbs/ul/cslib-marzena/informix5-ole
setenv DBDATE DMY4
setenv SQLEXEC $INFORMIXDIR/lib/sqlturbo
setenv TBCONFIG tbconfig.ole
setenv INFORMIXTERM termcap
setenv TERMCAP $INFORMIXDIR/etc/termcap
setenv TERM vt100

Man beachte, daß Informix seine eigene Termcap besitzt.

Im folgenden Text beschreibe ich die Entstehung, den Aufbau und die Besonderheiten eines Informix4 isql-Formulars. Ich werde dabei stärker auf die Elemente eingehen, die nicht ohne weiteres aus der Dokumentation zu ersehen sind.

Dieses Dokument soll und kann kein Manual sein oder dieser Aufgabe gerecht werden, sondern es ist konzipiert als ein Begleiter, auf dem unbequemen Weg ein Formular zu erzeugen.

1 Basis-Formularerzeugung

Isql unterstützt den Anwender bei der Erstellung von Formularen. Leider ist diese Unterstützung schon von vornherein insofern begrenzt, als daß sie vom Benutzer genaue Kenntnisse über die Syntax von Formularen erwartet, um ein brauchbares Formular zu erzeugen. Auch dann sind die Funktionen stark eingeschränkt. Es ist u.a. nicht möglich, auf Tabellenfeldern unterschiedlicher Tabellen gleichzeitig zu arbeiten, da in dem Formular immer erst auf die zu behandelnde Tabelle umgeschaltet werden muß.

1.1 Methoden

Es gibt drei verschiedene Möglichkeiten ein Formular zu erzeugen.

1. Zwei ärgerliche

(a) mit dem Formularbuilder ”sformbld” von Informix

(b) und innerhalb von isql über die Menüs

2. und eine umständliche von Hand.

1.1.1 sformbld

Um den Formularbuilder zu aktivieren, wird auf Komandozeilenebene das Komando ”sformbld -d” eingegeben. Daraufhin erscheint die Zeile:

Enter desired name for the form:

Der Name darf aber nicht mehr als zehn Buchstaben umfassen, da sonst, nachdem alle Tabellennamen von Hand eingegeben worden sind, die Fehlermeldung:

The form name ’’sformbloedie’’ exceeds the maximum length of 10 characters.
See error number -2040.
Exit -1

erscheint, was zur Folge hat, daß alles noch einmal eingegeben werden muß. Hat man diese Hürde genommen, wird mit:

Enter the name of the database used by the form:

nach der zu behandelnden Datenbank gefragt. Erweiternd wird darauf hingewiesen, daß sformbld nicht in der Lage ist, mehr als vierzehn Tabellen^1 in dem zu erzeugenden Formular aufzunehmen. Nach einer Leerzeile oder vierzehn Tabellen ist dann Schluß, und es erscheint die Meldung:

The form ’’sformbloed.per’’ will now be compiled.
The form compilation was successful.

Was der Benutzer erhält ist ein Standardformular mit allen Feldern^2 der Tabellen inclusive ihrer Namen, wobei die Formularfeldvariablen die ”sprechenden” Namen ”fnnn” bis ”fmmm” erhalten.

1.1.2 isql-Formulargenerator

Innerhalb von isql wird dem Benutzer unter dem Menüeintrag Form ein ”Generator” zur Verfügung gestellt. Wird er angewählt, erscheint eine Auswahl der vorhandenen Datenbanken, aus der wir mal cslibneu auswählen. Es folgt die Frage nach dem Namen des Formulars. Wird hier ein zu langer Name eingegeben, erhält man immerhin die Fehlermeldung ’’840: Name is too long’’ und darf den Namen nocheinmal eingeben. Danach werden die Tabellen der Datenbank aufgelistet. Wählt man eine aus, folgt die Frage, ob noch eine weitere ausgewählt werden soll oder nicht. Nur ist es diesmal nicht erlaubt, mehr als acht Tabellen^1 einzugeben. Eine Beschränkung, die einmal mehr die Mächtigkeit dieses Datenbanksystems zeigt.

1.1.3 Handarbeit

Wird alles von Hand eingegeben ist zu beachten, daß ein Formular mindestens die folgenden vier Abschnitte besitzen muß:

1. DATEBASE

2. SCREEN

3. TABLES

4. ATTRIBUTES

5. Optional existiert noch ein fünfter Abschnitt, der sich INSTRUCTIONS nennt.

Die Eintragungen in einem Formular zu diesen Abschnitten könnten folgendermaßen aussehen:

1cslibneu besitzt 37 Tabellen, und es werden warscheinlich noch mehr werden. Außerdem erlaubt einem isql Formulare mit mehr als vierzehn Tabellen.

2Die Formularfeldgrößen sind denen der Tabellen gleich, ist ein Feld länger als die Formularbreite, wird das Eingabefeld in mehrere Felder aufgeteilt.

database cslibneu
screen size 24 by 80
{
dokument [f000 ]
erscheinungsjahr [f001 ]
erscheinungsmonat [f002 ]
erscheinungstag [f003 ]
format [f004 ]
seiten [f005 ]
dokumentart [f006 ]
erfasst [f007 ]
geaendert [f008 ]
status [a]
}
end
tables
dok_dokument=cslib.dok_dokument
attributes
f000 = dok_dokument.dokument;
f001 = dok_dokument.erscheinungsjahr;
f002 = dok_dokument.erscheinungsmonat;
f003 = dok_dokument.erscheinungstag;
f004 = dok_dokument.format;
f005 = dok_dokument.seiten;
f006 = dok_dokument.dokumentart;
f007 = dok_dokument.erfasst;
f008 = dok_dokument.geaendert;
a = dok_dokument.status;
end

1.2 Zusammenfassung

Keine der vorangegangenen Methoden ist befriedigend, dennoch hat jede ein paar Vorteile. Ich habe daher eine Kombination der ”isql-Generate”-(siehe 1.1.2) und der ”Hand- Methode”(siehe 1.1.3) gewählt. Da mein Konferenzbandformular zwanzig Tabellen beinhaltet, wurden innerhalb von isql drei Formulare mit jeweils acht bzw. einmal vier Tabellen erzeugt, um alle Felder aller Tabellen als Basis zu besitzen. Danach ersetzte ich mit einem ”sed-Script^3” die Formularfeldnamen mit eigenen Namen, da sie doppelt und zudem nicht aussagekräftig waren. Daraufhin fügte ich die unter den vier Abschnitten (siehe 1.1.3) erzeugten Einträge in mein Konferenzbandformular-File ein. Ist der Rohbau des Formulars zur Zufriedenheit eingegeben, kann mit dem Komando:

/home/wbs/ul/cslib-marzena/informix5-ole/bin/sformbld sformbloed.per

das Formular compiliert werden. In isql kann man sich das Ergebnis unter dem Menü Form ansehen.

Jetzt konnte mit der Konfigurierung des Formulars begonnen werden. Die mehr oder

3So sieht es als File aus: s/f001/ tab var/g... und so von der Console: sed -e ’s/f001/ tab var/g’ usw. usw. bis fmmm.

weniger ”sprechenden” Variablennamen waren meiner Meinung nach notwendig, da es bei der Länge des Formulars (siehe 4) nicht einfach war, Fehler zu finden.

2 isql-Formularspezialitäten

In diesem Abschnitt werden die Eigenschaften eines isql-Formulars aufgeführt, die mir persönlich bei der Formularerzeugung besonders aufgefallen sind.

1. Es scheint keine Möglichkeit zu geben, Kommentare in den Formularquellcode einzubauen.

2. Damit das Formular aus der ”Auswahlliste” erscheint, rufe ich isql aus dem Verzeichnis auf, in dem mein übersetztes Formular steht.

3. Hat man hin und wieder so etwas wie: INFORMIX-SQL: 0%4;7m Form 0%4m auf dem Schirm, muß die Termcap neu gesetzt, also am besten das Environment neu ”gesourcet” werden.

4. Es ist möglich, Default-Werte in ein Formular einzugeben. Die Syntax dafür ist:

formular feld = tabellen feld, default = ’’wert’’;. Ganz besondere Default-

Werte sind ”today” bzw. ”current”. Diese Schlüsselworte erlauben es, das aktuelle Datum bzw. die aktuelle Uhrzeit einzutragen. Leider ist dies nur möglich, wenn das Feld der Tabelle vom Typ DATE ist. Diesem Anspruch genügt leider keines der Datumsfelder in cslibneu, so daß diese Funktion nicht benutzt werden kann.

5. Was mich anfangs verwunderte, war, daß sich der Cursor in meinem Formular kreuz und quer und nicht nach der Reihenfolge der Felder bewegte. Das lag daran, daß ein Formular nur eine Tabelle zur gleichen Zeit bearbeiten kann. Was bedeutet, man mus̈ sein Formular, will man nicht immer nach dem aktiven Feld suchen, so stricken, daß alle Felder einer Tabelle aufeinander folgen. Ihre Reihenfolge wird dann durch die Reihenfolge des Auftauchens der Felder in dem Abschnitt attributes bestimmt.

6. Die Größe eines Formularfeldes definiert sich durch den Raum zwischen den eckigen Klammern im Abschnitt screen. Dies führt zu der grotesken Eigenschaft, daß es nur 26 einstellige Felder geben kann, da das Alphabet nur 26 Buchstaben hat und isql verlangt, daß das erste Zeichen ein Buchstabe ist und Groß- und Kleinschreibung nicht unterschieden wird. Das bedeutet des weiteren, daß ”sprechende” Variablennamen u.U. zur Folge haben, daß Formularfelder größer als Tabellenfelder werden, da man ja mindestens den Formularfeldnamen zwischen die eckigen Klammern eintragen muß.

7. Sind Felder einer Tabelle größer definiert, als die Breite des Formulars es zuläßt, kann das Tabellenfeld in mehrere Formularfelder aufgeteilt werden. Die Syntax dafür lautet:

formular feld a = tabellen feld[n,m];
formular feld b = tabellen feld[m+1,k];

Durch das Kommando autonext ist es möglich, daß bei Erreichen des Feldendes automatisch in das nächste Feld übergesprungen wird. Die verbesserte Version würde dann so aussehen:

formular feld a = tabellen feld[n,m], autonext;

formular feld b = tabellen feld[m+1,l];

8. Besonderes Augenmerk ist auf die Möglichkeit zu richten, dem Formular anzugeben welche Tabellenfelder miteinander verknüpft werden sollen; eine für die Verwaltung der Haupt- und Nebenschlüssel wichtige Eigenschaft. Die Syntax dafür ist:

formular feld = tabelle1 feld a = tabelle2 feld b;. Es können dabei belie-

big viele Felder aneinandergehängt werden. Das Ergebnis dieser ”Gleichsetzung” ist, daß bei dem Eintragen in eines dieser Felder alle beschrieben werden.

9. Eine andere Möglichkeit Tabellen miteinander zu verknüpfen, ist die Funktion ”LOOKUP”. Hierbei wird nach Eingabe eines Wertes in ein Formularfeld in ein anderes Feld ein korrespondierender Wert aus einer anderen Tabelle eingetragen. Dies ist von Vorteil, wenn Werte wie ”1” in Wirklichkeit ”DIN A4” bedeuten. Die Syntax dafür ist:

formular feld = tabellen feld,
LOOKUP lookup formular feld = lookup tabellen anzeige feld
JOINING *lookup tabellen verknüpfungs feld;

wobei der ”*” angibt, daß der Eintrag mit einem der in der ”Lookup Tabelle” übereinstimmen muß; eine Funktion, die die Eingabe von Werten in ein Formular erheblich vereinfacht.

10. Während der Eingabe von Werten in ein Formular, das auf mehreren Tabellen arbeitet, ist es sehr umständlich, über das Menü von Tabelle zu Tabelle umzuschalten. Aus diesem Grund gibt es ”Hotkeys”. Möchte man von einer Tabelle zur nächsten, drückt man ¡t¿. Möchte man z.B. von Tabelle 1 zu Tabelle 4 wechseln, werden die Tasten ¡t¿ und ¡4¿ gedrückt. Die Nummern für die Tabellen ergeben sich aus dem Formular, durch die Reihenfolge des Auftauchens der Tabellen in dem Abschnitt

tables. Man scheint darauf zu vertrauen, daß der Benutzer nach ein paar dutzend

Eingaben die Nummern der zwanzig Tabellen im Kopf hat.

11. Um Queries zu vereinfachen, lassen sich Tabellenrelationen in dem Formular deklarieren^4.

Diese Eintragungen werden in dem Abschnitt instructions gemacht und sehen folgendermaßen aus: bla master of blub. Dadurch wird erreicht, daß bei einer Abfrage aus der Master-Tabelle heraus korrespondierende ”Rows” in den Detailtabellen gefunden werden.

3 Während der Eingabe

Das Eingeben von Daten kann auf vielfältige Weise geschehen. Ich gebe hier einfach mal vor, wie ich das gemacht habe.

In isql rufe ich mit ¡Form¿, ¡Run¿, ¡konferenz¿ mein Formular auf. Danach aktiviere

ich ¡Query¿ und trage in das Dokumentschlüsselfeld einen von mir ausgedachten Schlüssel

ein und schicke die Abfrage ab. Bekomme ich: There are no rows satisfying the

conditions kann ich den Schlüssel benutzen, wähle ¡Add¿ und trage diesen Schlüssel,

sowie die Felder der dok dokument-Tabelle ein. Besonderes Augenmerk ist darauf zu richten, daß man in der Lage ist Buchstaben mit ¡CTRL¿¡x¿ und das Ende der Zeile mit

4Da das in Informix nicht möglich ist

¡CTRL¿¡d¿ zu löschen. In den Feldern Erfasst: und Geaendert: wird ein Datum mit

folgendem Format erwünscht: ’’9999-09-09 09:09:09’’. Was der Status bedeutet, weiß ich nicht, aber es handelt sich um ein einstelliges Feld vom Typ char und ich habe schon einmal den Buchstaben ”F” darin gesehen. Ist alles zur Zufriedenheit gefüllt, drückt man

¡ESC¿ und erhält in der Ausgabezeile Row added. Jetzt wählt man die nächste Tabelle mit ¡t¿ aus, läßt den Dokumentschlüssel unberührt und wiederholt das Procedere bis zum

Schluß. Dabei ist noch zu beachten, das dok sprache und ko verantwortung übersprungen werden, da in ihnen nichts einzutragen ist.

Unangenehm einzutragen sind die Felder, die über eine ID einen Wert referenzieren, wie Dokumentart, Format, Titelart, Sprachen, Intervall und Verbindungsart. Ich habe dazu eine Tabelle gemacht (siehe Abschnitt: 5), die das erleichtert. Natürlich ist mit Hilfe der LOOKUP-Funktion (siehe Abschnitt: 2 Punkt 9) das Eintragen auch durch ausprobieren möglich.

Traut man dem Ganzen nicht, so kann man sich mit einer neuen Query der Daten versichern, indem man das Formular mit ¡e¿ beendet, es neu aufruft und eine Abfrage mit seinem Schlüssel startet. Danach läßt sich mit ¡Add¿ das Formular weiter füllen. Sollen in eine ”Detail”-Tabelle mehrere Datensätze eingetragen werden, ruft man unter demselben Hauptschlüssel der ”Master”-Tabelle die ”Detail”-Tabelle mit der ”¡Add¿,

¡ESC¿” Prozedur auf.

Anscheinend auf Grund der ”n zu m”-Beziehung von ko konferenz zu pr produzent über ko verantwortung kann das Formular Konferenzen und deren Produzenten nicht verknüpfen. Die jeweiligen Daten müssen daher seperat eingegeben und dann von Hand, durch das Eintragen der Nebenschlüssel in die Tabelle ko verantwortung, verknüpft werden.

3.1 Beispielformular

Das Eingabeformular für Konferenzbänd sieht wie Folgt aus:

-------------- K O N F E R E N Z B A N D E I N G A B E ---------------

============================ -1: DOK_DOKUMENT ========================
(Schluessel Dokument: [999999 ])
 Dokumentart:[7 ]=( Konferenzband )
Erscheinungsdatum:Jahr [9999 ] Monat [99 ] Tag[99 ]
Format:[9 ]=( 23,5x17,5 )Stat.:[F ]Seiten:[99 ]
Erfasst:[9999-09-09 09:09:09] Geaendert:[9999-09-09 09:09:09]
= -2- DOK_ISBN ======================| -3- DOK_ISSN ==================
| ISBN: 9999991 | ISSN: 9999993
============================ -4: DOK_TITEL ===========================
Titelart: 2 =( Titel )
Sprache: 1 =( Sprache nicht bekannt )
Titel: Dies ist ein Titel
       zweizeilig

============================ -5: DOK_TYP_F ===========================
Band: Keine Ahnung
============================ -7: KO_TAGUNG ===========================

(Schluessel Tagung: 9999994 )
Titel: Dies ist ein Tagungstitel
Titel:
Ort Tagungsort Der Ort
Tag: 99 Monat: 99 Jahr: 9999
============================ -8: KONFERENZ ===========================
(Schluessel Konferenz: 9999995 )
Titel: Konferenztitel
Intervall: 1 =( unbekannt )
Tag: 9999 Monat: 9999 Jahr: 9999
======================================================================

und für deren Produzenten zB. so:

                            P R O D U Z E N T
=========================== -10: PR_PRODUZENT ========================
(Schluessel Produzent: [9999996 ])
Name: [Produzentenname ]
      [ ]
Kurz: [kurzer Produzentenname ]
=========================== -11: PR_PERSON ===========================
Name: Personen-Nachname
Vorname Personen-Vorname
Titel Personen-Titel
=========================== -12: PR_INSTITUTION ======================
Namenszusatz: Namenszusatz der institution

Adr.: Adresse der institution

4 Konferenzbandformular (Souce-code)

database cslibneu
screen size 50 by 70
{
-------------- K O N F E R E N Z B A N D E I N G A B E ---------------

============================ -1: DOK_DOKUMENT ========================
(Schluessel Dokument: [_dokID ])
 Dokumentart:[_dok_art]=([_dok_artl ])
Erscheinungsdatum:Jahr [_dok_ejahr] Monat [_dok_emon] Tag[_dok_etag]
Format:[_dok_form]=([_dok_forml])Stat.:[_dok_stat]Seiten:[_dok_seiten]
Erfasst:[_dok_erfas ] Geaendert:[_dok_and ]
= -2- DOK_ISBN ======================| -3- DOK_ISSN ==================
| ISBN: [_isbn ] | ISSN: [_issn ]
============================ -4: DOK_TITEL ===========================
Titelart: [_dok_ti_tart]=([_dok_ti_tartl ])
Sprache: [_dok_ti_spr]=([_dok_ti_sprl ])
Titel:[_dok_ti_txt1 ]
      [_dok_ti_txt2 ]
      [_dok_ti_txt3 ]
============================ -5: DOK_TYP_F ===========================
Band: [_tf_bnd ]
============================ -7: KO_TAGUNG ===========================
(Schluessel Tagung: [_ko_t_ID])
Titel:[_ko_t_tit1 ]
Titel:[_ko_t_tit2 ]
Ort [_ko_t_ort ]
Tag: [_ko_t_tag] Monat: [_ko_t_mon] Jahr: [_ko_t_jahr]
============================ -8: KONFERENZ ===========================
(Schluessel Konferenz:[_ko_koID ])
Titel:[_ko_ko_tit ]
Intervall:[_ko_ko_int]=([_ko_ko_intl ])
Tag: [_ko_ko_tag] Monat: [_ko_ko_mon] Jahr: [_ko_ko_jahr]
======================================================================
                            P R O D U Z E N T
=========================== -10: PR_PRODUZENT ========================
(Schluessel Produzent: [_pr_prodID ])
Name: [_pr_prod_nam1 ]
      [_pr_prod_nam2 ]
Kurz: [_pr_prod_kurz ]
=========================== -11: PR_PERSON ===========================
Name: [_pr_per_nam ]
Vorname[_pr_per_vnam ]
Titel [_pr_per_tit ]
=========================== -12: PR_INSTITUTION ======================
Namenszusatz:[_pr_ins_namz1 ]
             [_pr_ins_namz2 ]
Adr.: [_pr_ins_adr1 ]
      [_pr_ins_adr2 ]
      [_pr_ins_adr3 ]
=========================== -13: PR_VERBINDUNG =======================
Verbindungsart: [_pr_verb_vart]=([_pr_verb_vartl])
(Sekundaerschluessel produzent_2:[_pr_verb_prod2] QUI BONO ?
}
end

tables
   dm_format=cslib.dm_format
   dm_titelart=cslib.dm_titelart
   dm_dokumentart=cslib.dm_dokumentart
   dm_intervall=cslib.dm_intervall
   dm_sprache=cslib.dm_sprache

   dok_dokument=cslib.dok_dokument
   dok_isbn=cslib.dok_isbn
   dok_issn=cslib.dok_issn
   dok_titel=cslib.dok_titel
   dok_sprache=cslib.dok_sprache
   dok_typ_f=cslib.dok_typ_f

   pr_produzent=cslib.pr_produzent
   pr_person=cslib.pr_person
   pr_institution=cslib.pr_institution
   pr_verbindung=cslib.pr_verbindung
   pr_rolle=cslib.pr_rolle
   pr_verbindungsart=cslib.pr_verbindungsart

   ko_tagung=cslib.ko_tagung
   ko_konferenz=cslib.ko_konferenz
   ko_verantwortung=cslib.ko_verantwortung

attributes
_dokID = dok_dokument.dokument = dok_isbn.dokument =
                   dok_issn.dokument = dok_titel.dokument =
                   dok_typ_f.dokument = dok_sprache.dokument,
                   color = red reverse;
_dok_art = dok_dokument.dokumentart,
                   LOOKUP _dok_artl = dm_dokumentart.text
                   JOINING *dm_dokumentart.dokumentart,
                   color=red underline;
_dok_ejahr = dok_dokument.erscheinungsjahr, color=red underline;
_dok_emon = dok_dokument.erscheinungsmonat, color=red underline;
_dok_etag = dok_dokument.erscheinungstag, color=red underline;

                                                                                    _dok_form = dok_dokument.format,
                                                                                                        LOOKUP _dok_forml = dm_format.text
                                                                                                        JOINING *dm_format.format,
                                                                                                        color=red underline;
                                                                                    _dok_stat = dok_dokument.status, color=red underline;
                                                                                    _dok_seiten = dok_dokument.seiten, color=red underline;
                                                                                    _dok_erfas = dok_dokument.erfasst, color=red underline;
                                                                                    _dok_and = dok_dokument.geaendert, color=red underline;
                                                                                    _isbn = dok_isbn.isbn, color=red underline;
                                                                                    _issn = dok_issn.issn, color=red underline;
                                                                                    _dok_ti_tart = dok_titel.titelart,
                                                                                                        LOOKUP _dok_ti_tartl = dm_titelart.text
                                                                                                        JOINING *dm_titelart.titelart,
                                                                                                        color=red underline;
                                                                                    _dok_ti_spr = dok_titel.sprache,
                                                                                                        LOOKUP _dok_ti_sprl = dm_sprache.text
                                                                                                        JOINING *dm_sprache.sprache,
                                                                                                        color=red underline;
                                                                                    _dok_ti_txt1 = dok_titel.text[1,62], color=red underline, autonext;
                                                                                    _dok_ti_txt2 = dok_titel.text[63,124], color=red underline, autonext;
                                                                                    _dok_ti_txt3 = dok_titel.text[125,186], color=red underline;
                                                                                    _tf_bnd = dok_typ_f.band, color=red underline;
                                                                                    _ko_t_ID = ko_tagung.tagung = dok_typ_f.tagung, color=red underline;
                                                                                    _ko_t_tit1 = ko_tagung.titel[1,62], color=red underline, autonext;
                                                                                    _ko_t_tit2 = ko_tagung.titel[63,100], color=red underline;
                                                                                    _ko_t_ort = ko_tagung.ort, color=red underline;
                                                                                    _ko_t_tag = ko_tagung.tag, color=red underline;
                                                                                    _ko_t_mon = ko_tagung.monat, color=red underline;
                                                                                    _ko_t_jahr = ko_tagung.jahr, color=red underline;
                                                                                    _ko_koID = ko_konferenz.konferenz =ko_tagung.konferenz
                                                                                                        =ko_verantwortung.konferenz,
                                                                                                        color=red underline;
                                                                                    _ko_ko_tit = ko_konferenz.titel, color=red underline;
                                                                                    _ko_ko_int = ko_konferenz.intervall,
                                                                                                        LOOKUP _ko_ko_intl = dm_intervall.text
                                                                                                        JOINING *dm_intervall.intervall,
                                                                                                        color=red underline;
                                                                                    _ko_ko_tag = ko_konferenz.tag, color=red underline;
                                                                                    _ko_ko_mon = ko_konferenz.monat, color=red underline;
                                                                                    _ko_ko_jahr = ko_konferenz.jahr, color=red underline;
                                                                                    _pr_prodID = pr_produzent.produzent =
                                                                                                        pr_person.produzent = pr_institution.produzent =
                                                                                                        pr_verbindung.produzent_1 =ko_verantwortung.produzent
                                                                                                        , color=red underline;
                                                                                    _pr_prod_nam1 = pr_produzent.name[1,62], color=red underline, autonext;
                                                                                    _pr_prod_nam2 = pr_produzent.name[63,124], color=red underline;
                                                                                    _pr_prod_kurz = pr_produzent.kurz, color=red underline;

                                                                                    _pr_per_nam = pr_person.name, color=red underline;
                                                                                    _pr_per_vnam = pr_person.vorname, color=red underline;
                                                                                    _pr_per_tit = pr_person.titel, color=red underline;

                                                                                    _pr_ins_namz1 = pr_institution.namenszusatz[1,55],
                                                                                                        color=red underline, autonext;
                                                                                    _pr_ins_namz2 = pr_institution.namenszusatz[56,100],
                                                                                                        color=red underline;
                                                                                    _pr_ins_adr1 = pr_institution.adresse[1,62],
                                                                                                        color=red underline, autonext;
                                                                                    _pr_ins_adr2 = pr_institution.adresse[63,124],
                                                                                                        color=red underline, autonext;
                                                                                    _pr_ins_adr3 = pr_institution.adresse[125,186],
                                                                                                        color=red underline;
                                                                                    _pr_verb_vart = pr_verbindung.verbindungsart,
                                                                                                        LOOKUP _pr_verb_vartl = pr_verbindungsart.text
                                                                                                        JOINING *pr_verbindungsart.verbindungsart,
                                                                                                        color=red underline;
                                                                                    _pr_verb_prod2 = pr_verbindung.produzent_2, color=red underline;

                                                                                    instructions
                                                                                    dok_dokument master of dok_titel;
                                                                                    dok_dokument master of dok_sprache;
                                                                                    dok_dokument master of dok_isbn;
                                                                                    dok_dokument master of dok_issn;
                                                                                    dok_dokument master of dok_typ_f;

                                                                                    ko_tagung master of dok_typ_f;

                                                                                    ko_konferenz master of ko_verantwortung;
                                                                                    ko_konferenz master of ko_tagung;

                                                                                    pr_produzent master of ko_verantwortung;
                                                                                    pr_produzent master of pr_person;
                                                                                    pr_produzent master of pr_institution;
                                                                                    pr_produzent master of pr_verbindung;

                                                                                    end

5 Konstanten

Diese Tabell gibt eine Übersicht, über die für die Formulareingabe relevanten Beziehungen zwischen ID-Feldern und deren Lookup-Feldern.

- - - Konstanten - - -

Typen ID Wert ID Wert

Tabelle: dm dokumentart DOKUMENTARTEN 1 unbekannt 10 ReviewSammlung 2 DOK TYP A 11 Sammelband 3 DOK TYP B 12 Sammelwerk 4 DOK TYP C 13 Zeitschrift 5 DOK TYP D 14 Set 6 FakyrDokument 15 FakyrDokument(91) 7 Konferenzband 16 FakyrDokument(92) 8 Monographie 17 FakyrDokument(93) 9 ReviewArtikel 18 FakyrDokument(94)

Tabelle: dm format FORMATE 1 unbekannt 7 A5 2 A4 8 20x14 3 A3 9 23,5x17,5 4 A2 10 virtuell 5 A1 11 25x19 6 A0

Tabelle: dm titelart TITELART 1 unbekannt 6 Zusatztitel 2 Titel 7 Fakyr-Titel 3 Untertitel 8 unbekannt 4 2.Untertitel 9 deutsch 5 3.Untertitel 10 englisch

Tabelle: dm sprache SPRACHEN 1 Sprache nicht bekannt

Tabelle: dm intervall INTERVALL keine Eintragungen

Tabelle: pr verbindungsart VERBINDUNGSART 1 Teil von

Tabelle 1: Konstanten